数理解析研究所講究録
第676巻 1988年 148-169

複合多項式とその計算法

愛知工業大学・電子工学科　秦野和郎　(Kazuo Hatano)

1・まえがき、

Fourier級数を有限項で打ち切ると両端、或は不連続点の付近で"Gibbsの現象"と呼ばれる振動を生ずる。従来これを避けるための方法として"Lanczosのσ-factorによる平滑化法"、"Fejer和による平滑化法"などが知られていたが、これらの方法には立ち上がりが緩やかになると言う欠点があった。又、満足できる程度に振動が減少する訳でもなかった。このように、Fourier級数を有限項で打ち切ることを前提とした方法では、どんなに工夫しても満足すべき結果は得られない筈である。

打ち切りFourier級数がFourier級数に一様収束しないことはよく知られている。実際の数値計算では近似式が元の関数に一様収束することは"絶対に必要"なことである。それにもかかわらずFourier級数を有限項で打ち切ることを前提とした近似手法がこれまで使われて来たのは、やむを得ないとはいえ不合理なことである。

Fourier級数は無限級数のままで使わなければならないのである。そのようにして使うことがFourier級数の正しい使い方であり、そのようにして初めてFourier級数が数値計算の有用な道具となるのである。無論、無限級数のままでは計算出来ない。幸いにも"普通に使う関数"のFourier級数展開式は本稿で述べるように"簡潔な形"で表現出来る。すなわちFourier級数、従って元の関数に急速に一様収束するような形にFourier級数を書き直すことができる。

本稿では"区分的に十分に滑らかな関数"のFourier級数展開式は実用上、有限級数化出来ることを示し、そのようにして書き直された式の計算法を述べる。

2・Fourier級数の簡潔な表現(複合多項式近似)、

以下では、まず十分に滑らかな関数のFourier展開式が簡潔な形に書き直し得ることについて述べる。次に区分的に十分に滑らかな関数のFourier展開式を簡潔な形に書き直すことが可能であることを示す。

2・1・十分に滑らかな関数の場合、

閉区間$[0,2\pi]$において与えられた関数、$f(x)$は以下の議論に必要な回数だけ微分可能であるとする。$f(x)$は

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{\infty}\{a_j(f)\cos jx+b_j(f)\sin jx\} \tag{2.1}$$

$$\begin{cases} a_j(f)=\dfrac{1}{\pi}\displaystyle\int_0^{2\pi}f(x)\cos jx\,dx & :0\leqq j \\ b_j(f)=\dfrac{1}{\pi}\displaystyle\int_0^{2\pi}f(x)\sin jx\,dx & :1\leqq j \end{cases} \tag{2.2}$$

とFourier展開される。式(2.2)の右辺に部分積分を反復適用すると

$$\begin{cases} a_j(f)=\displaystyle\sum_{i=1}^{m}\frac{(-1)^{i-1}}{j^{2i}}\omega_{2i-1}(f)+\frac{(-1)^{m+1}}{\pi j^{2m+1}}\int_0^{2\pi}f^{(2m+1)}(t)\sin jt\,dt \\ b_j(f)=\displaystyle\sum_{i=0}^{m-1}\frac{(-1)^{i-1}}{j^{2i+1}}\omega_{2i}(f)+\frac{(-1)^{m+1}}{\pi j^{2m+1}}\int_0^{2\pi}f^{(2m+1)}(t)(1-\cos jt)\,dt \end{cases} \tag{2.3}$$

となる。ここで

$$\omega_i(f)=\{f^{(i)}(2\pi)-f^{(i)}(0)\}/\pi \tag{2.4}$$

である。式(2.3)を参考にして次の量を定義する。

$$\begin{cases} a_j'(f)=\displaystyle\sum_{i=1}^{m}\frac{(-1)^{i-1}}{j^{2i}}\omega_{2i-1}(f) \\ b_j'(f)=\displaystyle\sum_{i=0}^{m-1}\frac{(-1)^{i-1}}{j^{2i+1}}\omega_{2i}(f) \end{cases} \tag{2.5}$$

式(2.3)と式(2.5)の差にSchwaltzの不等式を適用すると

$$\begin{cases} |a_j(f)-a_j'(f)|\leqq\dfrac{1}{\sqrt{\pi}\,j^{2m+1}}\cdot\|f^{(2m+1)}\|_2 \\ |b_j(f)-b_j'(f)|\leqq\dfrac{\sqrt{3}}{\sqrt{\pi}\,j^{2m+1}}\cdot\|f^{(2m+1)}\|_2 \end{cases} \tag{2.6}$$

となる。mが適当に大きければ、jが大きくなるにつれて式(2.6)の右辺は急速に小さくなる。従って、有限桁の計算では、あるJが存在して$j\geqq J$なる総てのjに対して

$$a_j(f)=a_j'(f)\quad,\quad b_j(f)=b_j'(f)\qquad j\geqq J \tag{2.7}$$

と見なし得るようになる。すなわち十分に大きなjに対してFourier係数は式(2.5)のように極めて"簡潔な形"で与えられる。

次に式(2.1)を書き直すと

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{j=n}^{\infty}\{a_j(f)\cos jx+b_j(f)\sin jx\} \tag{2.8}$$

であるが上式の最後の項に式(2.3)を代入すると式(2.1)は

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{i=1}^{m}\omega_{2i-1}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx$$
$$+\sum_{i=0}^{m-1}\omega_{2i}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx$$
$$+\frac{1}{\pi}\int_0^{2\pi}f^{(2m+1)}(t)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{m-1}}{j^{2m+1}}\{\sin jx-\sin j(x-t)\}dt \tag{2.9}$$

となる。上式を参考にして次の式を定義する。

$$H(f;x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{i=1}^{m}\omega_{2i-1}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx$$
$$+\sum_{i=0}^{m-1}\omega_{2i}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx \tag{2.10}$$

さて、式(2.9)と式(2.10)の差にSchwaltzの不等式を適用すると

$$|f(x)-H(f;x)|\leqq\{\frac{3}{\pi}\zeta(4m+2;n)\}^{1/2}\cdot\|f^{(2m+1)}\|_2 \tag{2.11}$$

となる。ここで

$$\zeta(x;n)=\sum_{j=0}^{\infty}\frac{1}{(j+x)^n} \tag{2.12}$$

はRiemannのZeta関数で

$$|\zeta(4m+2;n)|\leqq\frac{1}{(4m+1)n^{4m+1}}+\frac{1}{2\cdot n^{4m+2}}+O(n^{-4m-3}) \tag{2.13}$$

が成り立つ。従って

$$|f(x)-H(f;x)| \leqq \{\frac{3}{(4m+1)\pi}\}^{1/2} \cdot \frac{1}{n^{2m+1/2}} \cdot \|f^{(2m+1)}\|_2 \tag{2.14}$$

となる。上式で、nが十分に大きければその右辺は無視し得る程度に小さくなる。すなわち、Fourier級数、式(2.1)は

$$\begin{aligned} H(f;x)=&\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\} \\ &+\sum_{i=1}^{m}\frac{f^{(2i-1)}(2\pi)-f^{(2i-1)}(0)}{\pi}\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx \\ &+\sum_{i=0}^{m-1}\frac{f^{(2i)}(2\pi)-f^{(2i)}(0)}{\pi}\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx \end{aligned} \tag{2.15}$$

なる式で置き換えることが出来る。又、その誤差は$O(n^{-2m-1/2})$程度であることがわかる。

ここで、

$$\begin{cases} \tilde{q}_{2i}(x;n)=\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx \\ \tilde{q}_{2i+1}(x;n)=\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx \end{cases} \tag{2.16}$$

とおくとH(f;x)、従ってFourier級数は

$$\begin{aligned} H(f;x)=&\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\} \\ &+\sum_{i=1}^{2m}\omega_{i-1}(f)\cdot\tilde{q}_i(x;n) \end{aligned} \tag{2.17}$$

と簡潔に表現される(式中におけるnは十分に大きいとする)。

2・2・区分的に十分に滑らかな関数の場合、

次に、区分的に十分に滑らかな関数のFourier展開式も簡潔な形で表現出来る事を示す。

閉区間$[0,2\pi]$を

$$0=x_0<x_1<\ldots<x_\xi=2\pi \tag{2.18}$$

と分割したとき$f(x)$は各部分開区間(x_i,x_{i+1}) :$0\leqq i\leqq \xi-1$で以下の議論に必要な回数だけ微分可能であるとする。又、$f^{(\nu)}(x_i-)$, $f^{(\nu)}(x_i+)$:$0\leqq \nu\leqq m$, $0\leqq i\leqq \xi$が存在して

$$
\begin{cases}
f(x_i)=\{f(x_i-)+f(x_i+)\}/2 & :0\leqq i\leqq \xi \\
f^{(\nu)}(0+)=f^{(\nu)}(2\pi),\ f^{(\nu)}(0-)=f^{(\nu)}(2\pi-) & :0\leqq \nu\leqq m
\end{cases}
$$

であるとする。$f(x)$は

$$
f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{\infty}\{a_j(f)\cos jx+b_j(f)\sin jx\} \tag{2.19}
$$

$$
\begin{cases}
a_j(f)=\dfrac{1}{\pi}\displaystyle\int_0^{2\pi}f(x)\cos jx\,dx & :0\leqq j \\
b_j(f)=\dfrac{1}{\pi}\displaystyle\int_0^{2\pi}f(x)\sin jx\,dx & :1\leqq j
\end{cases} \tag{2.20}
$$

とFourier展開される。式(2.20)の右辺に部分積分を反復適用すると

$$
\begin{cases}
a_j(f)=\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\frac{(-1)^{\nu}}{j^{\nu}}\mathrm{cir}_{\nu}\,jx_i \\
\qquad\qquad +\dfrac{(-1)^m}{\pi j^{m+1}}\displaystyle\int_0^{2\pi}f^{(m+1)}(t)\,\mathrm{cir}_{m+1}\,jt\,dt \\
b_j(f)=-\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\frac{(-1)^{\nu}}{j^{\nu}}\mathrm{cir}_{\nu-1}\,jx_i \\
\qquad\qquad -\dfrac{(-1)^m}{\pi j^{m+1}}\displaystyle\int_0^{2\pi}f^{(m+1)}(t)\,\mathrm{cir}_{m}\,jt\,dt
\end{cases} \tag{2.21}
$$

となる。ここで

$$
\begin{cases}
\mathrm{cir}_{\nu}\,x=-\cos(x-\pi\nu/2) \\
\omega_{\nu}(f;x)=\{f^{(\nu)}(x-)-f^{(\nu)}(x+)\}/\pi
\end{cases} \tag{2.22}
$$

である。次に、

$$\begin{cases} a_j'(f)=\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}\omega_{\nu-1}(f;x_i)\frac{(-1)^\nu}{j^\nu}\mathrm{cir}_\nu jx_i \\ b_j'(f)=-\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}\omega_{\nu-1}(f;x_i)\frac{(-1)^\nu}{j^\nu}\mathrm{cir}_{\nu-1} jx_i \end{cases} \tag{2.23}$$

と置き、式(2.21)と式(2.23)との差にSchwaltzの不等式を適用すると、

$$\begin{cases} |a_j(f)-a_j'(f)|\leqq \dfrac{1}{j^{m+1}}\{\dfrac{2\xi}{\pi}\|f^{(m)}\|_\infty+\dfrac{1}{\sqrt{\pi}}\|f^{(m+1)}\|_2\} \\ |b_j(f)-b_j'(f)|\leqq \dfrac{1}{j^{m+1}}\{\dfrac{2\xi}{\pi}\|f^{(m)}\|_\infty+\dfrac{1}{\sqrt{\pi}}\|f^{(m+1)}\|_2\} \end{cases} \tag{2.24}$$

となる。mが適当に大きければ、上式から、jが大きくなるにつれて式(2.21)と式(2.23)との差は急速に小さくなる。すなわち、Fourier係数はjが十分に大きいとき式(2.23)のように簡潔に表現される。

さて、式(2.19)は

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{j=n}^{\infty}\{a_j(f)\cos jx+b_j(f)\sin jx\} \tag{2.25}$$

となるが、この式の最後の部分和に式(2.21)を代入すると、式(2.19)は

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\cdot\sum_{j=n}^{\infty}\frac{1}{j^\nu}\mathrm{cir}_\nu j(x-x_i)$$
$$-\frac{1}{\pi}\int_0^{2\pi}f^{(m+1)}(t)\cdot\sum_{j=n}^{\infty}\frac{1}{j^{m+1}}\mathrm{cir}_{m+1}j(x-t)dt \tag{2.26}$$

となる。式(2.16)を使って書けば

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$
$$+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\cdot\tilde{q}_\nu(x-x_i;n)$$
$$-\frac{1}{\pi}\int_0^{2\pi}f^{(m+1)}(t)\cdot\tilde{q}_{m+1}(x-t;n)dt$$

(2.27)

となる。ここでH(f;x)を

$$H(f;x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}$$

$$+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}\omega_{\nu-1}(f;x_i)\cdot\tilde{q}_\nu(x-x_i;n)$$

(2.28)

と定義すれば

$$f(x)-H(f;x)=\sum_{i=0}^{\xi-1}\omega_m(f;x_i)\cdot\tilde{q}_{m+1}(x-x_i;n)$$

$$-\frac{1}{\pi}\int_0^{2\pi}f^{(m+1)}(t)\cdot\tilde{q}_{m+1}(x-t;n)dt$$

(2.29)

となる。上式から

$$|f(x)-H(f;x)|\leqq\frac{2\xi}{\pi}\cdot\zeta(m+1;n)\cdot\|f^{(m)}\|_\infty+\{\zeta(2m+2;n)/\pi\}^{1/2}\|f^{(m+1)}\|_2$$

$$\leqq\frac{1}{n^m}\left[\frac{2\xi}{\pi m}\|f^{(m)}\|_\infty+\{\frac{1}{(2m+1)\pi n}\}^{1/2}\|f^{(m+1)}\|_2\right]$$

(2.30)

なる不等式を得ることが出来る。この式からmが適当に大きくnが十分に大きければFourier級数は式(2.28)のように簡潔な形で与えられる事がわかった。

以上のように"区分的に十分に滑らか"と言う程度の関数までならFourier級数は実用上、有限級数化出来ると言ってよい。

式(2.17)、式(2.28)をf(x)に対する"複合多項式近似"と呼ぶ。ここで、この名前の由来について述べる。

いま、

$$\frac{te^{xt}}{e^t-1}=\sum_{i=0}^{\infty}B_i(x)\frac{t^i}{i!}\quad:\ |t|<2\pi \tag{2.31}$$

で定義されるBernoulli多項式 $B_i(x)$を使って、多項式 $p_i(x)$を次のように定義する

。

$$\left\{\ p_1(x)=\begin{cases}(x-\pi)/2 & :\ x\in(0,2\pi)\\ 0 & :\ x=0\ ,\ x=2\pi\end{cases}\right.$$

$$\left\{ p_i(x) = \frac{(2\pi)^i}{2\cdot i!} B_i\left(\frac{x}{2\pi}\right) \ : \ i \geqq 2 \right. \tag{2.32}$$

この多項式は次のようにFourier展開される。

$$\begin{cases} p_{2i}(x) = \sum_{j=0}^{\infty} \dfrac{(-1)^{i-1}}{j^{2i}} \cos jx \\ p_{2i+1}(x) = \sum_{j=0}^{\infty} \dfrac{(-1)^{i-1}}{j^{2i+1}} \sin jx \end{cases} \tag{2.33}$$

従って式(2.16)で与えられる $\tilde{q}_i(x;n)$は、

$$\begin{cases} \tilde{q}_{2i}(x;n) = p_{2i}(x) - \sum_{j=1}^{n-1} \dfrac{(-1)^{i-1}}{j^{2i}} \cos jx \\ \tilde{q}_{2i+1}(x;n) = p_{2i+1}(x) - \sum_{j=1}^{n-1} \dfrac{(-1)^{i-1}}{j^{2i+1}} \sin jx \end{cases} \tag{2.34}$$

となる。すなわち、$\tilde{q}_i(x;n)$は i次の多項式とn-1次の三角多項式との和である。このことから式(2.17)で与えられるH(f;x)は、2m次の多項式とn-1次の三角多項式との和によるf(x)の近似であることがわかる。又、式(2.28)で与えられるH(f;x)は、m次の区分的多項式とn-1次の三角多項式との和によるf(x)の近似であることがわかる。

2・3・複合多項式近似についての補足、

式(2.17)、式(2.28)が複合多項式近似の、いわば標準形とも言うべき形であるがこれとは若干異なる形の式を導くこともできる。

式(2.17)に式(2.34)を代入すると、式(2.17)は

$$H(f;x) = \frac{1}{2} a_0(f) + \sum_{j=1}^{n-1} \{\tilde{a}_j(f)\cos jx + \tilde{b}_j(f)\sin jx\} + \sum_{i=1}^{2m} \omega_{i-1}(f)\cdot p_i(x) \tag{2.35}$$

$$\begin{cases} \tilde{a}_j(f) = a_j(f) - \sum_{i=1}^{m} \dfrac{(-1)^{i-1}}{j^{2i}} \omega_{2i-1}(f) = \dfrac{(-1)^{m+1}}{\pi j^{2m+1}} \int_0^{2\pi} f^{(2m+1)}(t) \sin jt\, dt \\ \tilde{b}_j(f) = b_j(f) - \sum_{i=0}^{m-1} \dfrac{(-1)^{i-1}}{j^{2i+1}} \omega_{2i}(f) = \dfrac{(-1)^{m+1}}{\pi j^{2m+1}} \int_0^{2\pi} f^{(2m+1)}(t)(1-\cos jt)\, dt \end{cases}$$

(2.36)

となる。式(2.35)は、n-1次の三角多項式と2m次の多項式の和を直接的に表している。その意味で直観的な表現であるが実際の計算には使えない。

式(2.36)に於ける$\omega_i(f)$は式(2.4)で定義されるように、両端のi次微係数の差であるから、その絶対値は、iが大きくなるにつれて急速に大きくなることが多い。このような場合、式(2.36)からわかるように小さなjに対する$\tilde{a}_j(f)$, $\tilde{b}_j(f)$の絶対値は$\|f\|_\infty$に比較して一般に極めて大きくなる。その結果、式(2.35)は桁落ちの激しい式になってしまうのである。

次に式(2.9)の最後の項を少し書き直すと式(2.9)は以下のようになる。

$$
\begin{aligned}
f(x)=&\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}\\
&+\sum_{i=1}^{m}\omega_{2i-1}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx\\
&+\sum_{i=0}^{m-1}\omega_{2i}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx\\
&+\sum_{j=n}^{\infty}\{\tilde{a}_j(f)\cos jx+\tilde{b}_j(f)\sin jx\}
\end{aligned}
\tag{2.37}
$$

$$
\left\{
\begin{aligned}
\tilde{a}_j(f)&=a_j(f)-\sum_{i=1}^{m}\frac{(-1)^{i-1}}{j^{2i}}\omega_{2i-1}(f)=\frac{(-1)^{m+1}}{\pi j^{2m+1}}\int_0^{2\pi}f^{(2m+1)}(t)\sin jt\,dt\\
\tilde{b}_j(f)&=b_j(f)-\sum_{i=0}^{m-1}\frac{(-1)^{i-1}}{j^{2i+1}}\omega_{2i}(f)=\frac{(-1)^{m+1}}{\pi j^{2m+1}}\int_0^{2\pi}f^{(2m+1)}(t)(1-\cos jt)\,dt
\end{aligned}
\right.
\tag{2.38}
$$

となる。この式から、H(f;x)を次のように定義してもよい筈である。

$$
\begin{aligned}
H(f;x)=&\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}\\
&+\sum_{i=1}^{m}\omega_{2i-1}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx\\
&+\sum_{i=0}^{m-1}\omega_{2i}(f)\cdot\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i+1}}\sin jx
\end{aligned}
$$

$$+\sum_{j=n}^{N}\{\tilde{a}_j(f)\cos jx+\tilde{b}_j(f)\sin jx\}$$

(2.39)

この式に於てnが適当に大きければ$\tilde{a}_j(f)$, $\tilde{b}_j(f)$の絶対値は$\|f\|_\infty$を越えることはなく、又、jの増大に対して$O(j^{-2m-1})$で減少する。

以上はf(x)が閉区間$[0,2\pi]$で十分に滑らかな関数の場合についてであった。

f(x)が区分的に十分に滑らかな場合についても同じような式を導くことができる。すなわち、式(2.28)は

$$H(f;x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{\tilde{a}_j(f)\cos jx+\tilde{b}_j(f)\sin jx\}$$
$$+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}\omega_{\nu-1}(f;x_i)\cdot\tilde{p}_\nu(x-x_i;n)$$

(2.40)

$$\begin{cases} a_j(f)=a_j(f)-\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\frac{(-1)^\nu}{j^\nu}\mathrm{cir}_\nu\, jx_i \\ \quad=\dfrac{(-1)^m}{\pi j^{m+1}}\displaystyle\int_0^{2\pi}f^{(m+1)}(t)\mathrm{cir}_{m+1}jt\,dt \\ b_j(f)=b_j(f)+\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\frac{(-1)^\nu}{j^\nu}\mathrm{cir}_{\nu-1}jx_i \\ \quad=-\dfrac{(-1)^m}{\pi j^{m+1}}\displaystyle\int_0^{2\pi}f^{(m+1)}(t)\mathrm{cir}_m jt\,dt \end{cases}$$

(2.41)

と書き直すことができる。ここで$\tilde{p}_\nu(x;n)$は$p_\nu(x;n)$を周期2πで実軸上に周期的に延長したものである。又、式(2.41)で定義される係数を使って

$$f(x)=\frac{1}{2}a_0(f)+\sum_{j=1}^{n-1}\{a_j(f)\cos jx+b_j(f)\sin jx\}+\sum_{j=n}^{\infty}\{\tilde{a}_j(f)\cos jx+\tilde{b}_j(f)\sin jx\}$$
$$+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m+1}\omega_{\nu-1}(f;x_i)\cdot\tilde{q}_\nu(x-x_i)$$

(2.42)

と書くことができる。

3・複合多項式の計算法、

ここでは複合多項式

$$h(x)=\frac{1}{2}a_0+\sum_{j=1}^{n-1}\{a_j\cos jx+b_j\sin jx\}+\sum_{i=1}^{2m}c_i(n)q_i(x;n) \tag{3.1}$$

$$q_i(x;n)=n^i\cdot\tilde{q}_i(x;n) \tag{3.2}$$

及び

$$h(x)=\frac{1}{2}a_0+\sum_{j=1}^{n-1}\{a_j\cos jx+b_j\sin jx\}+\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}c_{i,\nu}(n)q_\nu(x-x_i;n) \tag{3.3}$$

$$0=x_0<x_1<\dots<x_\xi<2\pi \tag{3.4}$$

の計算法について述べる。この際、任意のxに対する$h(x)$の値を求めるのは難しいので、ここでは閉区間$[0,2\pi]$上の等間隔離散点

$$\bar{x}_r=\frac{2\pi r}{N}\quad : \ r=0,1,\dots,N\quad (N>2n) \tag{3.5}$$

における$h(x)$の値を求めることを考える。但し、$N>2n$とする。

さて、

$$q_{2i}(\bar{x}_r;n)=n^{2i}\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos jx \tag{3.6}$$

に$x=\bar{x}_r$を代入すると

$$q_{2i}(x;n)=n^{2i}\sum_{j=n}^{\infty}\frac{(-1)^{i-1}}{j^{2i}}\cos\frac{2\pi jr}{N} \tag{3.7}$$

である。上式における$j=n,n+1,\dots$を次のように四通りに分ける。

$$\begin{cases} j & : \ j=n,n+1,\dots,N/2-1 \\ kN & : \ k=1,2,\dots \\ kN+j,kN-j & : \ j=1,2,\dots,N/2-1\ ;\ k=1,2,\dots \\ kN+N/2 & : \ k=0,1,\dots \end{cases} \tag{3.8}$$

それぞれについて各部分和を計算する。$\cos x$の周期性、対称性を使い

$$\begin{cases} \bar{\delta}_i(x)=\sum_{k=1}^{\infty}\left\{\frac{1}{(k+x)^i}+\frac{(-1)^i}{(k-x)^i}\right\} \\ \bar{\tau}_i(x)=1/x^i+\bar{\delta}_i(x) \end{cases} \tag{3.9}$$

とおくと

$$q_{2i}(\bar{x}_r;n)=(-1)^{i-1}\left(\frac{n}{N}\right)^{2i}\{\sum_{s=0}^{n-1(2)}\bar{\delta}_{2i}\left(\frac{s}{N}\right)\cos s\bar{x}_r$$

$$+\sum_{s=n}^{N/2(3)}\bar{\tau}_{2i}\left(\frac{s}{N}\right)\cos s\bar{x}_r\} \qquad (3.10)$$

となる。但し、ここで

$$\begin{cases} \sum_{s=0}^{n-1(2)}A(s)=\frac{1}{2}A(0)+\sum_{s=1}^{n-1}A(s) \\ \sum_{s=n}^{N/2(3)}A(s)=\sum_{s=n}^{N/2-1}A(s)+\frac{1}{2}A\left(\frac{N}{2}\right) \end{cases} \qquad (3.11)$$

である。

同じようにして

$$q_{2i+1}(\bar{x}_r;n)=(-1)^{i-1}\left(\frac{n}{N}\right)^{2i+1}\{\sum_{s=0}^{n-1}\bar{\delta}_{2i+1}\left(\frac{s}{N}\right)\sin s\bar{x}_r$$

$$+\sum_{s=n}^{N/2}\bar{\tau}_{2i+1}\left(\frac{s}{N}\right)\sin s\bar{x}_r\} \qquad (3.10)$$

を得ることができる。

ここで

$$q_1(0;n)=q_1(2\pi;n)=0 \qquad (3.11)$$

であるが、式(2.32),(2.34),(3.2)からわかるように

$$-q_1(0+;n)=q_1(2\pi-;n)=\pi n/2 \qquad (3.12)$$

である。

以上から、式(3.1)の形の複合多項式の等間隔離散点における値は

$$h(\bar{x}_r)=\frac{1}{2}\bar{u}_0+\sum_{j=1}^{N/2-1}\{\bar{u}_j\cos j\bar{x}_r+\bar{v}_j\sin j\bar{x}_r\}+\frac{1}{2}\bar{u}_{N/2}\cos\frac{N}{2}\bar{x}_r \qquad (3.13)$$

$$\begin{cases} \bar{u}_j=\begin{cases} a_j+\sum_{i=1}^{m}c_{2i}(n)(-1)^{i-1}\left(\frac{n}{N}\right)^{2i}\bar{\delta}_{2i}\left(\frac{j}{N}\right) & : 0\leqq j\leqq n-1 \\ \sum_{i=1}^{m}c_{2i}(n)(-1)^{i-1}\left(\frac{n}{N}\right)^{2i}\bar{\tau}_{2i}\left(\frac{j}{N}\right) & : n\leqq j\leqq N/2 \end{cases} \\ \bar{v}_j=\begin{cases} b_j+\sum_{i=0}^{m-1}c_{2i+1}(n)(-1)^{i-1}\left(\frac{n}{N}\right)^{2i+1}\bar{\delta}_{2i+1}\left(\frac{j}{N}\right) & : 1\leqq j\leqq n-1 \\ \sum_{i=0}^{m-1}c_{2i+1}(n)(-1)^{i-1}\left(\frac{n}{N}\right)^{2i+1}\bar{\tau}_{2i+1}\left(\frac{j}{N}\right) & : n\leqq j\leqq N/2-1 \end{cases} \end{cases} \qquad (3.14)$$

(3.15)

$$h(0+)=h(0)-c_1(n)\frac{\pi n}{2} \quad , \quad h(2\pi-;n)=h(2\pi)+c_1(n)\frac{\pi n}{2} \tag{3.16}$$

となる。すなわち、等間隔離散点におけるh(x)の値を計算するためには、まず式(3.14),(3.15)により離散Fourier係数、$\bar{u}_0,\bar{u}_j,\bar{v}_j:j=1,2,\ldots,N/2-1,\bar{u}_{N/2}$を計算する。次にそれらを使って実逆FFTを適用すれば$h(\bar{x}_r):r=0,1,\ldots,N/2$を一挙に計算することができる。

以上は式(3.1)の形の複合多項式、すなわち閉区間$[0,2\pi]$で連続な場合であった。式(3.3)で与えられる複合多項式、すなわち不連続点を持つ場合の計算法も同じようにして導くことができる。但し、式(3.4)で与えられる不連続点の位置については幾分厳しい制限が必要である。式(3.4)で与えられる$x_i:i=0,1,\ldots,\xi$について

$$x_i=\frac{2\pi r_i}{N} \quad : \; i=0,1,\ldots,\xi \tag{3.17}$$

とおいた時、r_iはすべて整数でなければならない。すなわち不連続点は等間隔離散点のどれかに一致しなければならない。

式(3.3)の形の複合多項式の等間隔離散点における値は

$$h(\bar{x}_r)=\frac{1}{2}\bar{u}_0+\sum_{j=1}^{N/2-1}\{\bar{u}_j\cos j\bar{x}_r+\bar{v}_j\sin j\bar{x}_r\}+\frac{1}{2}\bar{u}_{N/2}\cos\frac{N}{2}\bar{x}_r \tag{3.18}$$

$$\left\{\begin{array}{l}
\bar{u}_j=\left\{\begin{array}{ll}
a_j+\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}c_{i,\nu}(n)(-1)^{\nu}\left(\frac{n}{N}\right)^{\nu}\bar{\delta}_{\nu}\left(\frac{j}{N}\right)\mathrm{cir}_{\nu}\,jx_i : & 0\leqq j\leqq n-1 \\
\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}c_{i,\nu}(n)(-1)^{\nu}\left(\frac{n}{N}\right)^{\nu}\bar{\tau}_{\nu}\left(\frac{j}{N}\right)\mathrm{cir}_{\nu}\,jx_i : & n\leqq j\leqq N/2
\end{array}\right. \qquad (3.19) \\
\bar{v}_j=\left\{\begin{array}{ll}
b_j-\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}c_{i,\nu}(n)(-1)^{\nu}\left(\frac{n}{N}\right)^{\nu}\bar{\delta}_{\nu}\left(\frac{j}{N}\right)\mathrm{cir}_{\nu-1}\,jx_i & \\
\qquad : 1\leqq j\leqq n-1 & \\
-\displaystyle\sum_{i=0}^{\xi-1}\sum_{\nu=1}^{m}c_{i,\nu}(n)(-1)^{\nu}\left(\frac{n}{N}\right)^{\nu}\bar{\tau}_{\nu}\left(\frac{j}{N}\right)\mathrm{cir}_{\nu-1}\,jx_i & \\
\qquad : n\leqq j\leqq N/2-1 &
\end{array}\right. \qquad (3.20)
\end{array}\right.$$

$$h(x_i+)=h(x_i)-c_{i,1}(n)\cdot\frac{\pi n}{2} \quad , \quad h(x_i-;n)=h(x_i)+c_{i,1}(n)\cdot\frac{\pi n}{2} \tag{3.21}$$

で与えられる。

いずれの場合も式(3.9)で定義される$\bar{\delta}_i(x)$, $\bar{\tau}_i(x)$の計算を必要とする。これらの関数をここでは離散化関数と呼ぶことにする。

4・離散化関数の計算法、

次に離散化関数の計算法について述べる。まず、$\bar{\delta}_i(x):0\leqq x<1/2$の計算法を考える。

$$\bar{\delta}_i(x)=\sum_{k=1}^{K-1}\left\{\frac{1}{(k+x)^i}+\frac{(-1)^i}{(k-x)^i}\right\}+\sum_{k=K}^{\infty}\left\{\frac{1}{(k+x)^i}+\frac{(-1)^i}{(k-x)^i}\right\} \tag{4.1}$$

の右辺第一項はそのまま計算することとし、第二項をべき級数展開することにする。但し、$\bar{\delta}_{2i+1}(0)=0$であるからこの場合のみは因数、xをくくりだすことにする。

さて

$$g_i(x)=\frac{1}{(k+x)^i}+\frac{(-1)^i}{(k-x)^i} \tag{4.2}$$

とおく。この式に関して

$$g_i'(x)=-i\cdot g_{i+1}(x) \tag{4.3}$$

が成り立つ。すなわち、$g_i(x)$を微分すると$g_{i+1}(x)$の定数倍になる。これより

$$g_{i+1}(x)=-\frac{1}{i}g_i'(x) \tag{4.3}$$

が得られる。この式を反復適用すると

$$\begin{cases} g_{2i+1}(x)=\dfrac{1}{(2i)!}g_1^{(2i)}(x) \\ g_{2i}(x)=-\dfrac{1}{(2i-1)!}g_1^{(2i-1)}(x) \end{cases} \tag{4.4}$$

となる。次に

$$g_1(x)=\frac{1}{k+x}-\frac{1}{k-x} \tag{4.5}$$

において

$$\frac{1}{k-x}=\frac{1}{k}\cdot\frac{1}{1-x/k}=\frac{1}{k}\sum_{r=0}^{\infty}\left(\frac{x}{k}\right)^r \tag{4.6}$$

などを使うと

$$g_1(x)=-\frac{2}{k}\sum_{r=0}^{\infty}\left(\frac{x}{k}\right)^{2r+1} \tag{4.7}$$

となる。この式と式(4.4)とから

$$\begin{cases} g_{2i+1}(x)=-2\sum_{r=i+1}^{\infty}\binom{2r-1}{2i}\frac{1}{k^{2r}}x^{2r-2i-1} \\ g_{2i}(x)=2\sum_{r=i}^{\infty}\binom{2r-1}{2i-1}\frac{1}{k^{2r}}x^{2r-2i} \end{cases} \tag{4.8}$$

を得ることができる。又、

$$g_{2i+1}(x)=\frac{(-2x)}{(k^2-x^2)(k-x)^{2i}}\sum_{r=0}^{2i}\left(\frac{k-x}{k+x}\right)^r \tag{4.9}$$

である。以上を使うと、

$$\begin{cases} \bar{\delta}_{2i+1}(x)=-2x\left\{\sum_{k=1}^{K-1}\frac{1}{(k^2-x^2)(k-x)^{2i}}\sum_{r=0}^{2i}\left(\frac{k-x}{k+x}\right)^r \right. \\ \qquad\qquad \left. +\sum_{r=i+1}^{\infty}\binom{2r-1}{2i}\zeta(2r;K)x^{2r-2i-2}\right\} \\ \bar{\delta}_{2i}(x)=\sum_{k=1}^{K-1}\left\{\frac{1}{(k+x)^{2i}}+\frac{1}{(k-x)^{2i}}\right\} \\ \qquad\qquad +2\sum_{r=i}^{\infty}\binom{2r-1}{2i-1}\zeta(2r;K)x^{2r-2i} \end{cases} \tag{4.10}$$

を得ることができる。ここで

$$\zeta(r;K)=\sum_{j=0}^{\infty}\frac{1}{(j+K)^r} \tag{4.11}$$

は一般化されたRiemannのZeta関数である。

次に、y=1/2-xとおくと

$$\bar{\tau}_i(x)=\sum_{k=0}^{\infty}\left\{\frac{1}{(k+1/2-y)^i}+\frac{(-1)^i}{(k+1/2+y)^i}\right\} \tag{4.12}$$

となる。$\bar{\delta}_i(x)$の場合と同じようにして

$$\begin{cases} \bar{\tau}_{2i+1}(x)=(1-2x)\left\{\sum_{k=0}^{K-1}\frac{1}{(k+1-x)(k+x)^{2i+1}}\sum_{r=0}^{2i}\left(\frac{k+x}{k+1-x}\right)^r \right. \\ \qquad\qquad \left. +\sum_{r=i+1}^{\infty}\binom{2r-1}{2i}\zeta\left(2r;K+\frac{1}{2}\right)\left(\frac{1}{2}-x\right)^{2r-2i-2}\right\} \end{cases}$$

$$\left\{\begin{aligned} \bar{\tau}_{2i}(x) &= \sum_{k=0}^{K-1}\left\{\frac{1}{(k+x)^{2i}}+\frac{1}{(k+1-x)^{2i}}\right\} \\ &\qquad +2\sum_{r=i}^{\infty}\binom{2r-1}{2i-1}\zeta\left(2r;K+\frac{1}{2}\right)\left(\frac{1}{2}-x\right)^{2r-2i} \end{aligned}\right. \qquad (4.13)$$

を得ることができる。これらの式の級数部はKを適当にとれば収束は早い。

式(4.10)及び、式(4.13)における、べき級数の部分は漸化式

$$T_{r+1}(x)=2xT_r(x)-T_{r-1}(x) \quad , \ T_0(x)=1 \quad , \ T_1(x)=x \qquad (4.14)$$

で定義されるChebyshev多項式を使って

$$\sum_{r=0}^{\infty} d_r T_r(8x^2-1) \qquad (4.15)$$

の形に書き直すとよい。このようにするとK=5としたとき$\bar{\delta}_i(x)$, $\bar{\tau}_i(x)$:$1\leqq i\leqq 12$を10^{-17}程度の絶対誤差で計算するのに必要な項数は7程度である。

$0\leqq x\leqq 0.5$が与えられたとき$\bar{\delta}_i(x)$:$L\leqq i\leqq N(\leqq 12)$を計算するための SUBROUTINE DDELC1 及び$\bar{\tau}_i(x)$:$L\leqq i\leqq N(\leqq 12)$を計算するための SUBROUTINE DTAUC1 を付録に示す。

4・むすび、

普通に使う関数のFourier級数展開式は簡潔な形に書き直されることについて述べた。更に三角多項式と多項式との和、すなわち複合多項式の計算法について述べた。ここで述べたことは近い将来、実用計算に大いに使われるようになると思われる。

FFT(高速Fourier変換)は多くの分野で使われている。FFTは総ての離散Fourier係数を少ない計算量で求め、丸めの誤差を大幅に減らすという顕著な性質がある。しかし、離散Fourier係数の打ち切り誤差については少しも改善されない。本稿で述べた複合多項式をうまく使えば、離散Fourier係数の打ち切り誤差を大幅に改善し得る筈である。このような検討をも含めて今後に残された課題は多い。

```
C*
C********* DDELC1 ** DDELC1A1 ******************************
C*                                                          *
C*     CALCULATION OF DELTA<I>(X) ; I=LDEG,...,NDEG         *
C*      0.0<=(X=IS/NL)=<0.5  ;  1<=LDEG<=NDEG<=12           *
C*                                                          *
C*********            **           ************************
      SUBROUTINE DDELC1(X,DELTA,LDEG,NDEG,ICON)
      IMPLICIT INTEGER*4 (I-N)
      IMPLICIT REAL    *8 (A-H,O-Z)
      DIMENSION DELTA(12),DL(84),FTERM(12),STERM(12)
     1          ,AMUP1(12),AMUM1(12)
      DATA K,KLM1 / 5 , 4 / , KSTEM / 7 /
C --- THE SECOND TERM OF DELTA< 1>   NTERM= 7 -------------
      DATA DL( 1),DL( 2),DL( 3),DL( 4),DL( 5),DL( 6),DL( 7)
     1/  2.21771 78053 99450D-01 , 4.49631 53630 50485D-04
     2,  8.08435 85362 41811D-07 , 1.70623 92760 50478D-09
     3,  3.87014 24084 60299D-12 , 9.12383 25319 88305D-15
     4,  2.20018 45095 59229D-17 /
C --- THE SECOND TERM OF DELTA< 2>   NTERM= 7 -------------
      DATA DL( 8),DL( 9),DL(10),DL(11),DL(12),DL(13),DL(14)
     1/  2.22674 28762 44579D-01 , 1.35538 26327 20736D-03
     2,  4.06271 62447 18252D-06 , 1.20057 80217 09444D-08
     3,  3.50142 87881 87037D-11 , 1.00891 71293 97819D-13
     4,  2.87534 79185 90518D-16 /
C --- THE SECOND TERM OF DELTA< 3>   NTERM= 7 -------------
      DATA DL(15),DL(16),DL(17),DL(18),DL(19),DL(20),DL(21)
     1/  1.08433 49204 52681D-02 , 1.30009 16077 30121D-04
     2,  5.76285 52184 84390D-07 , 2.24094 20280 70144D-09
     3,  8.07142 79059 65610D-12 , 2.76036 30440 80393D-14
     4,  9.08707 83056 02405D-17 /
C --- THE SECOND TERM OF DELTA< 4>   NTERM= 7 -------------
      DATA DL(22),DL(23),DL(24),DL(25),DL(26),DL(27),DL(28)
     1/  3.70189 53928 86985D-03 , 1.31554 92916 50897D-04
     2,  9.69482 87022 90668D-07 , 5.27209 71015 51666D-09
     3,  2.43990 37607 40950D-11 , 1.01942 99819 02615D-13
     4,  3.96493 70277 49647D-16 /
C --- THE SECOND TERM OF DELTA< 5>   NTERM= 7 -------------
      DATA DL(29),DL(30),DL(31),DL(32),DL(33),DL(34),DL(35)
     1/  5.26282 98386 44781D-04 , 1.55125 06711 90053D-05
     2,  1.26534 40823 90304D-07 , 7.80788 23546 56802D-10
     3,  4.07780 17903 97061D-12 , 1.90320 28576 16195D-14
     4,  8.18627 86602 78047D-17 /
C --- THE SECOND TERM OF DELTA< 6>   NTERM= 7 -------------
      DATA DL(36),DL(37),DL(38),DL(39),DL(40),DL(41),DL(42)
     1/  1.11563 77049 28739D-04 , 9.51184 60975 76533D-06
     2,  1.28421 42549 27939D-07 , 1.10622 90182 97803D-09
     3,  7.41656 61392 95249D-12 , 4.22652 65143 44734D-14
     4,  2.14704 14536 49329D-16 /
C --- THE SECOND TERM OF DELTA< 7>   NTERM= 7 -------------
      DATA DL(43),DL(44),DL(45),DL(46),DL(47),DL(48),DL(49)
     1/  2.53737 73322 57269D-05 , 1.36998 67655 38158D-06
     2,  1.77007 91403 86266D-08 , 1.58226 94835 70385D-10
     3,  1.12711 10978 06801D-12 , 6.87071 87398 35709D-15
     4,  3.73606 48205 57290D-17 /
C --- THE SECOND TERM OF DELTA< 8>   NTERM= 7 -------------
      DATA DL(50),DL(51),DL(52),DL(53),DL(54),DL(55),DL(56)
     1/  4.02650 12095 71645D-06 , 6.07640 50333 13563D-07
     2,  1.29172 64641 91876D-08 , 1.60822 96176 18288D-10
     3,  1.46890 23069 83484D-12 , 1.09256 85709 23677D-14
```

```
           4,  7.01323 85171 89695D-17 /
      C --- THE SECOND TERM OF DELTA< 9>   NTERM= 7  -------------
            DATA DL(57),DL(58),DL(59),DL(60),DL(61),DL(62),DL(63)
           1/  1.21624 60536 95779D-06 , 1.03361 62125 55069D-07
           2,  1.93009 40661 32998D-09 , 2.35041 20156 79287D-11
           3,  2.18524 99105 27230D-13 , 1.68324 50571 21232D-15
           4,  1.12768 67987 65696D-17 /
      C --- THE SECOND TERM OF DELTA<10>   NTERM= 7  -------------
            DATA DL(64),DL(65),DL(66),DL(67),DL(68),DL(69),DL(70)
           1/  1.58981 38470 67692D-07 , 3.62012 44008 30885D-08
           2,  1.10400 55677 62236D-09 , 1.86732 42045 68573D-11
           3,  2.22295 81958 36043D-13 , 2.08758 34737 60096D-15
           4,  1.65011 20849 23171D-17 /
      C --- THE SECOND TERM OF DELTA<11>   NTERM= 7  -------------
            DATA DL(71),DL(72),DL(73),DL(74),DL(75),DL(76),DL(77)
           1/  5.80116 38677 06213D-08 , 7.06848 13370 08789D-09
           2,  1.79296 52753 59325D-10 , 2.84570 33304 77598D-12
           3,  3.34038 97349 47313D-14 , 3.16839 80746 20046D-16
           4,  2.56176 93115 97410D-18 /
      C --- THE SECOND TERM OF DELTA<12>   NTERM= 7  -------------
            DATA DL(78),DL(79),DL(80),DL(81),DL(82),DL(83),DL(84)
           1/  6.62573 92828 89303D-09 , 2.06131 86847 66510D-09
           2,  8.46519 94436 13121D-11 , 1.86007 13150 35700D-12
           3,  2.79121 70054 83677D-14 , 3.22476 01734 43920D-16
           4,  3.07444 05707 90074D-18 /
      C +++ CONDITION CODE +++++++++++++++++++++++++++++++++++++
            ICON=0
            IF(LDEG.LE.0) ICON=-1
            IF(LDEG.GT.NDEG) ICON=-2
            IF(NDEG.GT.12) ICON=-3
            IF(X.LT.-0.00000 1D0) ICON=-11
            IF(X.GT. 0.50000 1D0) ICON=-12
            IF(ICON.LT.0) RETURN
      C +++ THE SECOND TERM ++++++++++++++++++++++++++++++++++++
            X1=8.0D0*X**2-1.0D0
            X2=2.0D0*X1
            DO 1110 I1=LDEG,NDEG
              ISTA=KSTEM*(I1-1)+1
              IEND=KSTEM*I1
              ISTD=ISTA+IEND
              BNP2=0.0D0
              BNP1=0.0D0
              DO 1100 IR1D=ISTA,IEND
                IR1=ISTD-IR1D
                BN=X2*BNP1-BNP2+DL(IR1)
                IF(IR1.NE.ISTA) THEN
                  BNP2=BNP1
                  BNP1=BN
                ENDIF
       1100   CONTINUE
              SUM1=BN-X1*BNP1
              STERM(I1)=SUM1
       1110 CONTINUE
      C +++ THE FIRST TERM +++++++++++++++++++++++++++++++++++++
            DO 1120 I1=LDEG,NDEG
              FTERM(I1)=0.0D0
       1120 CONTINUE
            DO 1160 KS1D=1,KLM1
              KS1=K-KS1D
              AK1=FLOAT(KS1)
```

Fri Oct 07 13:04:52 1988   DDELC1A1.FOR

```
            AKPXI=1.0D0
            AKMXI=1.0D0
            AK1PX=1.0D0/(AK1+X)
            AK1MX=1.0D0/(AK1-X)
            DO 1130 I1=1,NDEG
              AKPXI=AKPXI*AK1PX
              AKMXI=AKMXI*AK1MX
              AMUP1(I1)=AKPXI
              AMUM1(I1)=AKMXI
 1130       CONTINUE
            OK1X=1.0D0/(AK1**2-X**2)
            OK2X=(AK1-X)/(AK1+X)
            DO 1150 I1=LDEG,NDEG
              IF(MOD(I1,2).NE.0) THEN
                SUM2=0.0D0
                DO 1140 IR1=1,I1
                  SUM2=SUM2*OK2X+1.0D0
 1140           CONTINUE
                SUM2=SUM2*OK1X
                IF(I1.NE.1) THEN
                  SUM2=SUM2*AMUM1(I1-1)
                ENDIF
                FTERM(I1)=FTERM(I1)+SUM2
              ELSE
                FTERM(I1)=FTERM(I1)+AMUP1(I1)+AMUM1(I1)
              ENDIF
 1150       CONTINUE
 1160 CONTINUE
C +++ DELTA<I> ++++++++++++++++++++++++++++++++++++++++++
      XM2=X*2.0D0
      DO 1170 I1=LDEG,NDEG
        IF(MOD(I1,2).NE.0) THEN
          DELTA(I1)=-XM2*(FTERM(I1)+STERM(I1))
        ELSE
          DELTA(I1)=FTERM(I1)+2.0D0*STERM(I1)
        ENDIF
 1170 CONTINUE
      RETURN
      END
```

Fri Oct 07 13:05:22 1988   DTAUC1A1.FOR

```
C*
C********* DTAUC1 ** DTAUC1A1 *******************************
C*                                                          *
C*      CALCULATION OF TAU<I>(X) ; I=LDEG,...,NDEG          *
C*       0.0<=(X=IS/NL)=<0.5   ;  1<=LDEG<=NDEG<=12         *
C*                                                          *
C*********          **          *****************************
      SUBROUTINE DTAUC1(X,TAU,LDEG,NDEG,ICON)
      IMPLICIT INTEGER*4 (I-N)
      IMPLICIT REAL   *8 (A-H,O-Z)
      DIMENSION TAU(12),TU(84),FTERM(12),STERM(12)
     1          ,AMUP1(12),AMUM1(12)
      DATA K / 5  / , KSTEM / 7 /
C --- THE SECOND TERM OF TAU< 1>   NTERM= 7  ---------------
      DATA TU( 1),TU( 2),TU( 3),TU( 4),TU( 5),TU( 6),TU( 7)
     1/  1.99670 71205 10727D-01 , 3.28805 67573 45720D-04
     2,  4.81442 20210 58213D-07 , 8.29448 62526 74963D-10
     3,  1.53897 92410 13451D-12 , 2.97313 44093 80299D-15
```

```
     4,  5.88410 52169 48709D-18 /
C --- THE SECOND TERM OF TAU< 2>   NTERM= 7  ---------------
      DATA TU( 8),TU( 9),TU(10),TU(11),TU(12),TU(13),TU(14)
     1/  2.00330 25416 03837D-01 , 9.90278 54288 73379D-04
     2,  2.41718 90773 04411D-06 , 5.83082 36489 67249D-09
     3,  1.39104 17407 68086D-11 , 3.28460 28875 37377D-14
     4,  7.68252 14804 03789D-17 /
C --- THE SECOND TERM OF TAU< 3>   NTERM= 7  ---------------
      DATA TU(15),TU(16),TU(17),TU(18),TU(19),TU(20),TU(21)
     1/  7.92236 82841 80130D-03 , 7.73509 40747 83052D-05
     2,  2.79882 16285 26890D-07 , 8.90274 08936 46456D-10
     3,  2.62770 22610 10740D-12 , 7.37527 30707 76466D-15
     4,  1.99509 80838 25359D-17 /
C --- THE SECOND TERM OF TAU< 4>   NTERM= 7  ---------------
      DATA TU(22),TU(23),TU(24),TU(25),TU(26),TU(27),TU(28)
     1/  2.69273 16856 89116D-03 , 7.81008 68342 20246D-05
     2,  4.70045 43151 92516D-07 , 2.09136 99491 62186D-09
     3,  7.93243 53689 88380D-12 , 2.72027 66743 83183D-14
     4,  8.69452 21244 52519D-17 /
C --- THE SECOND TERM OF TAU< 5>   NTERM= 7  ---------------
      DATA TU(29),TU(30),TU(31),TU(32),TU(33),TU(34),TU(35)
     1/  3.12428 57035 22625D-04 , 7.52098 07464 13252D-06
     2,  5.01939 66905 32899D-08 , 2.53842 10522 73803D-10
     3,  1.08812 54365 15040D-12 , 4.17341 97521 28020D-15
     4,  1.47667 61766 69131D-17 /
C --- THE SECOND TERM OF TAU< 6>   NTERM= 7  ---------------
      DATA TU(36),TU(37),TU(38),TU(39),TU(40),TU(41),TU(42)
     1/  6.55345 67902 45146D-05 , 4.59351 15147 15256D-06
     2,  5.08066 86724 10716D-08 , 3.58877 71355 07869D-10
     3,  1.97539 06213 33435D-12 , 9.25268 00525 33114D-15
     4,  3.86697 21964 74963D-17 /
C --- THE SECOND TERM OF TAU< 7>   NTERM= 7  ---------------
      DATA TU(43),TU(44),TU(45),TU(46),TU(47),TU(48),TU(49)
     1/  1.22522 35184 32093D-05 , 5.41980 13462 78520D-07
     2,  5.74229 01604 87572D-09 , 4.21429 04042 28462D-11
     3,  2.46743 67498 17403D-13 , 1.23745 38379 95272D-15
     4,  5.54024 75239 40338D-18 /
C --- THE SECOND TERM OF TAU< 8>   NTERM= 7  ---------------
      DATA TU(50),TU(51),TU(52),TU(53),TU(54),TU(55),TU(56)
     1/  1.90848 84940 04796D-06 , 2.38912 63039 96351D-07
     2,  4.17444 83474 89581D-09 , 4.27104 44253 96878D-11
     3,  3.20796 53670 24214D-13 , 1.96365 67913 32774D-15
     4,  1.03803 71231 15453D-17 /
C --- THE SECOND TERM OF TAU< 9>   NTERM= 7  ---------------
      DATA TU(57),TU(58),TU(59),TU(60),TU(61),TU(62),TU(63)
     1/  4.78081 54310 20496D-07 , 3.34007 19773 63961D-08
     2,  5.12564 60555 87798D-10 , 5.13299 37229 66401D-12
     3,  3.92745 11154 48947D-14 , 2.49135 72765 85916D-16
     4,  1.37532 78339 79323D-18 /
C --- THE SECOND TERM OF TAU<10>   NTERM= 7  ---------------
      DATA TU(64),TU(65),TU(66),TU(67),TU(68),TU(69),TU(70)
     1/  6.07738 17304 21646D-08 , 1.15961 00610 90639D-08
     2,  2.91672 48450 69215D-10 , 4.06270 71281 73582D-12
     3,  3.98318 34968 44527D-14 , 3.08188 08629 72123D-16
     4,  2.00790 73618 47788D-18 /
C --- THE SECOND TERM OF TAU<11>   NTERM= 7  ---------------
      DATA TU(71),TU(72),TU(73),TU(74),TU(75),TU(76),TU(77)
     1/  1.85732 64437 29866D-08 , 1.86721 37868 85228D-09
     2,  3.90069 19696 88021D-11 , 5.09886 04093 08742D-13
     3,  4.93126 64138 20222D-15 , 3.85533 34774 76823D-17
```

```
     4,  2.57033 06482 49344D-19 /
C --- THE SECOND TERM OF TAU<12>   NTERM= 7  ---------------
      DATA TU(78),TU(79),TU(80),TU(81),TU(82),TU(83),TU(84)
     1/  2.04243 80768 06599D-09 , 5.38172 27529 66039D-10
     2,  1.82939 00774 21065D-11 , 3.31716 34793 28862D-13
     3,  4.10533 41680 80669D-15 , 3.91180 10139 31388D-17
     4,  3.07642 29972 07158D-19 /
C +++ CONDITION CODE ++++++++++++++++++++++++++++++++++++++
      ICON=0
      IF(LDEG.LE.0) ICON=-1
      IF(LDEG.GT.NDEG) ICON=-2
      IF(NDEG.GT.12) ICON=-3
      IF(X.LT.-0.00000 1D0) ICON=-11
      IF(X.GT. 0.50000 1D0) ICON=-12
      IF(ICON.LT.0) RETURN
C +++ THE SECOND TERM +++++++++++++++++++++++++++++++++++++
      X1=8.0D0*(0.5D0-X)**2-1.0D0
      X2=2.0D0*X1
      DO 1110 I1=LDEG,NDEG
        ISTA=KSTEM*(I1-1)+1
        IEND=KSTEM*I1
        ISTD=ISTA+IEND
        BNP2=0.0D0
        BNP1=0.0D0
        DO 1100 IR1D=ISTA,IEND
          IR1=ISTD-IR1D
          BN=X2*BNP1-BNP2+TU(IR1)
          IF(IR1.NE.ISTA) THEN
            BNP2=BNP1
            BNP1=BN
          ENDIF
 1100   CONTINUE
        SUM1=BN-X1*BNP1
        STERM(I1)=SUM1
 1110 CONTINUE
C +++ THE FIRST TERM ++++++++++++++++++++++++++++++++++++++
      DO 1120 I1=LDEG,NDEG
        FTERM(I1)=0.0D0
 1120 CONTINUE
      DO 1160 KS1D=1,K
        KS1=K-KS1D
        AK1=FLOAT(KS1)
        AKPXI=1.0D0
        AKMXI=1.0D0
        AK1PX=1.0D0/(AK1+X)
        AK1MX=1.0D0/(AK1+1.0D0-X)
        DO 1130 I1=1,NDEG
          AKPXI=AKPXI*AK1PX
          AKMXI=AKMXI*AK1MX
          AMUP1(I1)=AKPXI
          AMUM1(I1)=AKMXI
 1130   CONTINUE
        OK1X=1.0D0/((AK1+X)*(AK1+1.0D0-X))
        OK2X=(AK1+X)/(AK1+1.0D0-X)
        DO 1150 I1=LDEG,NDEG
          IF(MOD(I1,2).NE.0) THEN
            SUM2=0.0D0
            DO 1140 IR1=1,I1
              SUM2=SUM2*OK2X+1.0D0
 1140       CONTINUE
```

```
            SUM2=SUM2*OK1X
            IF(I1.NE.1) THEN
              SUM2=SUM2*AMUP1(I1-1)
            ENDIF
            FTERM(I1)=FTERM(I1)+SUM2
          ELSE
            FTERM(I1)=FTERM(I1)+AMUP1(I1)+AMUM1(I1)
          ENDIF
 1150   CONTINUE
 1160 CONTINUE
C +++ TAU<I> +++++++++++++++++++++++++++++++++++++++++++++++++
      XM2=1.0D0-X*2.0D0
      DO 1170 I1=LDEG,NDEG
        IF(MOD(I1,2).NE.0) THEN
          TAU(I1)=XM2*(FTERM(I1)+STERM(I1))
        ELSE
          TAU(I1)=FTERM(I1)+2.0D0*STERM(I1)
        ENDIF
 1170 CONTINUE
      RETURN
      END
```